

NAG Fortran Library Routine Document

D01BCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D01BCF returns the weights (normal or adjusted) and abscissae for a Gaussian integration rule with a specified number of abscissae. Six different types of Gauss rule are allowed.

2 Specification

```
SUBROUTINE D01BCF(ITYPE, A, B, C, D, N, WEIGHT, ABSCIS, IFAIL)
INTEGER          ITYPE, N, IFAIL
real           A, B, C, D, WEIGHT(N), ABSCIS(N)
```

3 Description

This routine returns the weights w_i and abscissae x_i for use in the summation

$$S = \sum_{i=1}^n w_i f(x_i)$$

which approximates a definite integral (see Davis and Rabinowitz (1975), or Stroud and Secrest (1966)). The following types are provided:

(a) Gauss–Legendre

$$S \simeq \int_a^b f(x) dx, \quad \text{exact for } f(x) = P_{2n-1}(x).$$

Constraint: $b > a$.

(b) Gauss–Jacobi

normal weights:

$$S \simeq \int_a^b (b-x)^c (x-a)^d f(x) dx, \quad \text{exact for } f(x) = P_{2n-1}(x),$$

adjusted weights:

$$S \simeq \int_a^b f(x) dx, \quad \text{exact for } f(x) = (b-x)^c (x-a)^d P_{2n-1}(x).$$

Constraint: $c > -1$, $d > -1$, $b > a$.

(c) Gauss–Exponential

normal weights:

$$S \simeq \int_a^b \left| x - \frac{a+b}{2} \right|^c f(x) dx, \quad \text{exact for } f(x) = P_{2n-1}(x),$$

adjusted weights:

$$S \simeq \int_a^b f(x) dx, \quad \text{exact for } f(x) = \left| x - \frac{a+b}{2} \right|^c P_{2n-1}(x).$$

Constraint: $c > -1$, $b > a$.

(d) Gauss–Laguerre

normal weights:

$$\begin{aligned}
 S &\simeq \int_a^\infty |x-a|^c e^{-bx} f(x) dx \quad (b > 0), \\
 &\simeq \int_{-\infty}^a |x-a|^c e^{-bx} f(x) dx \quad (b < 0), \quad \text{exact for } f(x) = P_{2n-1}(x),
 \end{aligned}$$

adjusted weights:

$$\begin{aligned}
 S &\simeq \int_a^\infty f(x) dx \quad (b > 0), \\
 &\simeq \int_{-\infty}^a f(x) dx \quad (b < 0), \quad \text{exact for } f(x) = |x-a|^c e^{-bx} P_{2n-1}(x).
 \end{aligned}$$

Constraint: $c > -1$, $b \neq 0$.

(e) Gauss–Hermite

normal weights:

$$S \simeq \int_{-\infty}^{+\infty} |x-a|^c e^{-b(x-a)^2} f(x) dx, \quad \text{exact for } f(x) = P_{2n-1}(x),$$

adjusted weights:

$$S \simeq \int_{-\infty}^{+\infty} f(x) dx, \quad \text{exact for } f(x) = |x-a|^c e^{-b(x-a)^2} P_{2n-1}(x).$$

Constraint: $c > -1$, $b > 0$.

(f) Gauss–Rational

normal weights:

$$\begin{aligned}
 S &\simeq \int_a^\infty \frac{|x-a|^c}{|x+b|^d} f(x) dx \quad (a+b > 0), \\
 &\simeq \int_{-\infty}^a \frac{|x-a|^c}{|x+b|^d} f(x) dx \quad (a+b < 0), \quad \text{exact for } f(x) = P_{2n-1}\left(\frac{1}{x+b}\right),
 \end{aligned}$$

adjusted weights:

$$\begin{aligned}
 S &\simeq \int_a^\infty f(x) dx \quad (a+b > 0), \\
 &\simeq \int_{-\infty}^a f(x) dx \quad (a+b < 0), \quad \text{exact for } f(x) = \frac{|x-a|^c}{|x+b|^d} P_{2n-1}\left(\frac{1}{x+b}\right).
 \end{aligned}$$

Constraint: $c > -1$, $d > c+1$, $a+b \neq 0$.

In the above formulae, $P_{2n-1}(x)$ stands for any polynomial of degree $2n-1$ or less in x .

The method used to calculate the abscissae involves finding the eigenvalues of the appropriate tridiagonal matrix (see Golub and Welsch (1969)). The weights are then determined by the formula

$$w_i = \left\{ \sum_{j=0}^{n-1} P_j^*(x_i)^2 \right\}^{-1}$$

where $P_j^*(x)$ is the j th orthogonal polynomial with respect to the weight function over the appropriate interval.

The weights and abscissae produced by D01BCF may be passed to D01FBE, which will evaluate the summations in one or more dimensions.

4 References

Davis P J and Rabinowitz P (1975) *Methods of Numerical Integration* Academic Press

Stroud A H and Secrest D (1966) *Gaussian Quadrature Formulas* Prentice-Hall

Golub G H and Welsch J H (1969) Calculation of Gauss quadrature rules *Math. Comput.* **23** 221–230

5 Parameters

1: ITYPE – INTEGER *Input*

On entry: indicates the type of quadrature rule.

If ITYPE = 0, Gauss–Legendre.

If ITYPE = 1, Gauss–Jacobi.

If ITYPE = 2, Gauss–Exponential.

If ITYPE = 3, Gauss–Laguerre.

If ITYPE = 4, Gauss–Hermite.

If ITYPE = 5, Gauss–Rational.

The above values give the normal weights; the adjusted weights are obtained if the value of ITYPE above is negated.

Constraint: $-5 \leq \text{ITYPE} \leq 5$.

2: A – *real* *Input*

3: B – *real* *Input*

4: C – *real* *Input*

5: D – *real* *Input*

On entry: the parameters a , b , c and d which occur in the quadrature formulae. C is not used if ITYPE = 0; D is not used unless ITYPE = ± 1 or ± 5 . For some rules C and D must not be too large (See Section 6.)

Constraints:

if ITYPE = 0, $A < B$;

if ITYPE = ± 1 , $A < B$, $C > -1$ and $D > -1$;

if ITYPE = ± 2 , $A < B$, and $C > -1$;

if ITYPE = ± 3 , $B \neq 0$, and $C > -1$;

if ITYPE = ± 4 , $B > 0$, and $C > -1$;

if ITYPE = ± 5 , $A + B \neq 0$, $C > -1$ and $D > C + 1$.

6: N – INTEGER *Input*

On entry: the number of weights and abscissae to be returned, n . If ITYPE = -2 or -4 and $C \neq 0.0$, an odd value of N may raise problems – see Section 6, IFAIL = 6.

Constraint: $N > 0$.

7: WEIGHT(N) – *real* array *Output*

On exit: the N weights.

8: ABSCIS(N) – *real* array *Output*

On exit: the N abscissae.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

The algorithm for computing eigenvalues of a tridiagonal matrix has failed to obtain convergence. If the soft fail option is used, the values of the weights and abscissae on return are indeterminate.

$IFAIL = 2$

On entry, $N < 1$,
or $ITYPE < -5$,
or $ITYPE > 5$.

If the soft fail option is used, weights and abscissae are returned as zero.

$IFAIL = 3$

A , B , C or D is not in the allowed range:

if $ITYPE = 0$, $A \geq B$;
if $ITYPE = \pm 1$, $A \geq B$ or $C \leq -1.0$ or $D \leq -1.0$ or $C + D + 2.0 > GMAX$;
if $ITYPE = \pm 2$, $A \geq B$ or $C \leq -1.0$;
if $ITYPE = \pm 3$, $B = 0.0$ or $C \leq -1.0$ or $C + 1.0 > GMAX$;
if $ITYPE = \pm 4$, $B \leq 0.0$ or $C \leq -1.0$ or $(C + 1.0/2.0) > GMAX$;
if $ITYPE = \pm 5$, $A + B = 0.0$ or $C \leq -1.0$ or $D \leq C + 1.0$.

Here $GMAX$ is the (machine-dependent) largest integer value such that $\Gamma(GMAX)$ can be computed without overflow (see the Users' Note for your implementation for S14AAF).

If the soft fail option is used, weights and abscissae are returned as zero.

$IFAIL = 4$

One or more of the weights are larger than $RMAX$, the largest floating-point number on this machine. $RMAX$ is given by the function X02ALF. If the soft fail option is used, the overflowing weights are returned as $RMAX$. Possible solutions are to use a smaller value of N ; or, if using adjusted weights, to change to normal weights.

$IFAIL = 5$

One or more of the weights are too small to be distinguished from zero on this machine. If the soft fail option is used, the underflowing weights are returned as zero, which may be a usable approximation. Possible solutions are to use a smaller value of N ; or, if using normal weights, to change to adjusted weights.

$IFAIL = 6$

Gauss-Exponential or Gauss-Hermite adjusted weights with N odd and $C \neq 0.0$. Theoretically, in these cases:

for $C > 0.0$, the central adjusted weight is infinite, and the exact function $f(x)$ is zero at the central abscissa.

for $C < 0.0$, the central adjusted weight is zero, and the exact function $f(x)$ is infinite at the central abscissa.

In either case, the contribution of the central abscissa to the summation is indeterminate.

In practice, the central weight may not have overflowed or underflowed, if there is sufficient rounding error in the value of the central abscissa.

If the soft fail option is used, the weights and abscissa returned may be usable; the user must be particularly careful not to ‘round’ the central abscissa to its true value without simultaneously ‘rounding’ the central weight to zero or ∞ as appropriate, or the summation will suffer. It would be preferable to use normal weights, if possible.

Note: remember that, when switching from normal weights to adjusted weights or vice versa, redefinition of $f(x)$ is involved.

7 Accuracy

The accuracy depends mainly on n , with increasing loss of accuracy for larger values of n . Typically, one or two decimal digits may be lost from machine accuracy with $n \simeq 20$, and three or four decimal digits may be lost for $n \simeq 100$.

8 Further Comments

The major portion of the time is taken up during the calculation of the eigenvalues of the appropriate tridiagonal matrix, where the time is roughly proportional to n^3 .

9 Example

This example program returns the abscissae and (adjusted) weights for the seven-point Gauss–Laguerre formula.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      D01BCF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N
      PARAMETER       (N=7)
      INTEGER          NOUT
      PARAMETER       (NOUT=6)
*      .. Local Scalars ..
      real            A, B, C, D
      INTEGER          IFAIL, ITYPE, J
*      .. Local Arrays ..
      real            ABSCIS(N), WEIGHT(N)
*      .. External Subroutines ..
      EXTERNAL        D01BCF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'D01BCF Example Program Results'
      A = 0.0e0
      B = 1.0e0
      C = 0.0e0
      D = 0.0e0
      ITYPE = -3
      IFAIL = 0

*
      CALL D01BCF(ITYPE,A,B,C,D,N,WEIGHT,ABSCIS,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Laguerre formula,', N, ' points'
      WRITE (NOUT,*)
```

```
      WRITE (NOUT,*) '      Abcissae          Weights'
      WRITE (NOUT,*)
      WRITE (NOUT,99998) (ABSCIS(J),WEIGHT(J),J=1,N)
      STOP
*
99999 FORMAT (1X,A,I3,A)
99998 FORMAT (1X,e15.5,5X,e15.5)
      END
```

9.2 Program Data

None.

9.3 Program Results

D01BCF Example Program Results

Laguerre formula, 7 points

Abcissae	Weights
0.19304E+00	0.49648E+00
0.10267E+01	0.11776E+01
0.25679E+01	0.19182E+01
0.49004E+01	0.27718E+01
0.81822E+01	0.38412E+01
0.12734E+02	0.53807E+01
0.19396E+02	0.84054E+01
